

**NASA
Technical
Memorandum**

NASA TM- 100337

**MODELING, DESIGNING, AND SIMULATING A
POINTING CONTROL SYSTEM FOR BALLOON-BORNE
SOLAR EXPERIMENTS**

By W. D. Lightsey and M. E. Polites

Structures and Dynamics Laboratory
Science and Engineering Directorate

July 1988

(NASA-TM-100337) MODELING, DESIGNING AND
SIMULATING A POINTING CONTROL SYSTEM FOR
BALLOON-BORNE SOLAR EXPERIMENTS (NASA)
39 p CSCL 14B

N88-28300

Unclas
G3/35 0161156



National Aeronautics and
Space Administration

George C. Marshall Space Flight Center

TABLE OF CONTENTS

	Page
INTRODUCTION	1
MODELING THE SYSTEM.....	2
DERIVING THE EQUATIONS OF MOTION.....	3
MODELING THE GIMBAL BEARINGS.....	11
DEVELOPING THE CONTROL LAW	13
SIMULATION RESULTS	14
CONCLUSIONS	15
APPENDIX A	21
APPENDIX B	23
APPENDIX C	24
APPENDIX D	26
REFERENCES	32

PRECEDING PAGE BLANK NOT FILMED

LIST OF SYMBOLS

<u>Symbol</u>	<u>Definition</u>
$()_T$	a quantity subscripted by T pertains to the entire system
$()_P$	a quantity subscripted by P pertains to the instrument (body P)
$()_G$	a quantity subscripted by G pertains to the gondola (body G)
$(\dot{\ })$	a dot over a quantity denotes its time derivative
$(\bar{\ })$	an overbar denotes a vector quantity
g	the acceleration due to gravity
m	mass
W	weight
\bar{i}_{ab}	unit vectors - the subscript a denotes the coordinate frame and the subscript b denotes the axis direction, $b = 1, 2, 3$
θ_{CE}	coarse elevation angle traversed by the coarse elevation gimbal axis
θ_{ab}	Euler angles - subscript a denotes the coordinate frame and subscript b denotes the axis about which the angle is measured, $b = 1, 2, 3$
Ω_{ab}	magnitude of the Euler angle rates (first time derivatives of θ_{ab})
\bar{a}_B	acceleration of the balloon attach-point
\bar{r}_P	position vector of the center-of-mass of body P
\bar{r}_G	position vector of the center-of-mass of body G
\bar{r}_{CM}	position vector of the center-of-mass of the system
\bar{f}_C	force in the cable $\bar{f}_C = f_C \bar{i}_{C1}$
\bar{l}_C	vector representing the cable $\bar{l}_C = l_C \bar{i}_{C1}$
$\bar{f}_{G(g)}$	force on body G due to gravity (weight of G), $\bar{f}_{G(g)} = -W_G \bar{i}_{LS1}$
$\bar{f}_{P(g)}$	force on body P due to gravity, $\bar{f}_{P(g)} = -W_P \bar{i}_{LS1}$
$\bar{\rho}_G$	vector from origin of C frame to center-of-mass of body G, $\bar{\rho}_G = -\rho_G \bar{i}_{G1}$
$\bar{\rho}_P$	vector from origin of C frame to center-of-mass of body P, $\bar{\rho}_P = -\rho_P \bar{i}_{G1}$
$\bar{\rho}_{CM}$	vector from origin of C frame to center-of-mass of the system, $\bar{\rho}_{CM} = -\rho_{CM} \bar{i}_{G1}$

V PRECEDING PAGE BLANK NOT FILMED

ρ_{G-P}	denotes the scalar quantity $\rho_G - \rho_P$
$\Sigma \bar{F}_{\text{ext}}$	sum of external forces acting on the system
$\Sigma \bar{M}_{\text{CM}}()$	sum of moments about the center-of-mass of a body (the character in the subscript parenthesis denotes the body)
$H_{\text{CM}}()$	angular momentum about the center-of-mass of a body
$\Sigma \bar{F}(P)$	sum of the forces acting on body P
$\bar{\omega}_P$	angular rate of body P, $\bar{\omega}_P = \omega_{P1} \bar{i}_{P1} + \omega_{P2} \bar{i}_{P2} + \omega_{P3} \bar{i}_{P3}$
$\bar{\omega}_G$	angular rate of body G, $\bar{\omega}_G = \omega_{G1} \bar{i}_{G1} + \omega_{G2} \bar{i}_{G2} + \omega_{G3} \bar{i}_{G3}$
$\bar{T}_{P/G}$	torque on body P due to body G, (note $\bar{T}_{P/G} = -\bar{T}_{G/P}$), $\bar{T}_{P/G} = T_{P/G(P1)} \bar{i}_{P1} + T_{P/G(P2)} \bar{i}_{P2} + T_{P/G(P3)} \bar{i}_{P3}$
$\bar{T}_{G/C}$	torque on gondola due to cable torsion, $\bar{T}_{G/C} = -K_{G/C} \theta_{G1} \bar{i}_{G1}$
$K_{G/C}$	torsional spring constant of the cable
$\bar{f}_{P/G}$	force on body P due to body G, (note $\bar{f}_{P/G} = -\bar{f}_{G/P}$), $\bar{f}_{P/G} = f_{P/G(\text{LS1})} \bar{i}_{(\text{LS1})} + f_{P/G(\text{LS2})} \bar{i}_{\text{LS2}} + f_{P/G(\text{LS3})} \bar{i}_{\text{LS3}}$
T_A	output torque of the azimuth torque motor
T_E	output torque of the fine elevation torque motor
T_R	output torque of the roll torque motor
T_{fA}	resistance torque due to azimuth bearing friction
T_{fE}	resistance torque due to fine elevation axis flex pivot
T_{fR}	resistance torque due to roll axis flex pivot
T'_A	azimuth torque less bearing friction
T'_E	elevation torque less flex pivot stiffness effects
T'_R	roll torque less flex pivot stiffness effects
\bar{X}	denotes vector of state variables
\bar{U}	denotes vector of input variables
$\theta_i, i=1,2,3$	pointing angles – Euler angles between the P and S reference frames
$K_{Ri}, i=1,2,3$	rate gain in the control law
$K_{Pi}, i=1,2,3$	proportional gain in the control law

K_{ii} , $i=1,2,3$ integral control gain in the control law

I_P inertia tensor for the P body

$$I_P = \begin{bmatrix} I_{P11} & I_{P12} & I_{P13} \\ I_{P21} & I_{P22} & I_{P23} \\ I_{P31} & I_{P32} & I_{P33} \end{bmatrix}$$

I_G inertia tensor for the G body

$$I_G = \begin{bmatrix} I_{G11} & I_{G12} & I_{G13} \\ I_{G21} & I_{G22} & I_{G23} \\ I_{G31} & I_{G32} & I_{G33} \end{bmatrix}$$

TECHNICAL MEMORANDUM

MODELING, DESIGNING, AND SIMULATING A POINTING CONTROL SYSTEM FOR BALLOON-BORNE SOLAR EXPERIMENTS

INTRODUCTION

The objective of this effort is to mathematically model a balloon-borne pointing system for solar instruments and use the math model to develop a computer simulation of the system. The pointing system is one proposed by the Space Sciences Division of the Perkin-Elmer Corporation [1]. The concept is shown in Figure 1.

The gimbal system provides four angular motions: azimuth, fine elevation, coarse elevation, and fine roll (cross-elevation). The azimuth bearing allows unlimited, 360-deg rotation. Flexure pivots (bearings) are used on the fine elevation and roll axes and provide about ± 3.5 deg of rotation. The coarse elevation axis provides 80 deg of rotation.




Figure 1

A universal joint is located between the gondola and the balloon suspension cable to reduce bending moments between the gondola and the cable.

Using the gondola and gimbal system configuration outlined above, a math model of the system can now be developed.

MODELING THE SYSTEM

The system is modeled as two rigid bodies connected together by a gimbal system. Figure 2 shows the simplified model and the coordinate systems used to derive the equations of motion. The instrument and its counter-weight form body P, which is connected through the gimbal system to the gondola structure (body G). The gondola is suspended from the balloon-attach-point by a cable.




Figure 2

The LS, S, and B coordinate systems all have their origins fixed at the balloon-attach-point. The LS1 axis points away from the Earth along the local vertical. The LS3 axis is directed so that the Sun lies in the LS1 to LS3 plane. The S1 axis points to the Sun and the S2 axis is aligned with LS2.

The B coordinate system is fixed in the balloon and rotates with it. Only balloon rotation about LS1 is considered, so B1 remains aligned with LS1 as the balloon rotates.

The C coordinate frame has its origin fixed at the universal joint where the cable and gondola connect. The C1 axis remains aligned with the cable. The origin of the G coordinate frame is at the center-of-mass of the gondola and the G1 axis points along the vertical pylon of the gondola as shown. The origin of the P coordinate system is fixed at the center-of-mass of the P body. The P1 axis is aligned with the line-of-sight of the instrument.

Any friction in the balloon-cable and cable-gondola connections is neglected, as is any internal cable friction. The dynamics of the balloon are limited to a translational acceleration, \bar{a}_B , applied to the balloon-attach-point, and an angular rotation about the B1 axis.

Several other general assumptions apply to the model. Any aerodynamic disturbances on bodies P and G are neglected. The cable has a torsional stiffness, but is assumed to be massless and extensionless. The center-of-mass of each body lies along the G1 axis. Finally, it is assumed that the dynamics of the cable, body P, and body G do not affect the balloon motion.

DERIVING THE EQUATIONS OF MOTION

The orientation of the system at any time can be described by the Euler angles between the coordinate systems. Two angles are needed to express the transformation from the B frame to the C frame, θ_{c2} and θ_{c3} . Three angles are required from the C frame to the G frame, θ_{G1} , θ_{G2} , and θ_{G3} . Finally, θ_{P1} , θ_{P2} , and θ_{P3} define the transformation from the G frame to the P frame. Eight equations of motion, containing these angles and time derivatives of these angles, are required to express the motion of the system relative to the B coordinate frame.

Several more assumptions are made in order to simplify the equations. First, it is assumed that the Euler angles will be small, so that $\cos \theta \approx 1$ and $\sin \theta \approx \theta$. Also, higher order terms in products of state variables and/or inputs are neglected. During the initial derivation of the equations, the azimuth bearing friction and the flex pivot stiffnesses are neglected. The friction and stiffnesses will be modeled and included in the equations later in the derivation.

These equations can be derived in four basic steps. The first step is to solve the equation

$$\Sigma \bar{F}_{ext} = m_T (\ddot{\bar{r}}_{CM} + \bar{a}_B) \quad (1)$$

$\ddot{\bar{r}}_{CM}$ is defined relative to the LS coordinate frame. Since this frame accelerates with respect to inertial space, the addition of the term \bar{a}_B is necessary. The external forces acting on the G body and the P body are the cable force, \bar{f}_C , the gondola weight, $\bar{f}_{G(g)}$, and the instrument weight, $\bar{f}_{P(g)}$; therefore,

$$\Sigma \bar{F}_{ext} = \bar{f}_C - \bar{f}_{G(g)} - \bar{f}_{P(g)} \quad (2)$$

In terms of unit vectors, equation (2) can be written as

$$\Sigma \bar{F}_{ext} = f_C \bar{i}_{C1} - (W_P + W_G) \bar{i}_{LS1} \quad (3)$$

The vector \bar{r}_{CM} can be expressed in terms of the physical dimensions λ_C and ρ_{CM} as

$$\bar{r}_{CM} = -\lambda_C \bar{i}_{C1} - \rho_{CM} \bar{i}_{G1} \quad (4)$$

From this expression $\ddot{\bar{r}}_{CM}$ can be obtained. The acceleration of the balloon-attach-point, \bar{a}_B , can be expressed as

$$\bar{a}_B = a_{B(LS1)} \bar{i}_{LS1} + a_{B(LS2)} \bar{i}_{LS2} + a_{B(LS3)} \bar{i}_{LS3} \quad (5)$$

Notice that the components of equations (3), (4), and (5) are expressed in different coordinate systems. Coordinate transformations are needed to express all of the components in a common coordinate frame. Transforming all of the quantities in the above equations to the LS frame, and substituting the results into equation (1), the following equations result.

$$f_C = W_T + m_T a_{B(LS1)} \quad (6a)$$

$$-m_T (\rho_{CM} + \lambda_C) \ddot{\theta}_{C3} - m_T \rho_{CM} \ddot{\theta}_{G3} = (W_T + m_T a_{B(LS1)}) \theta_{C3} - m_T a_{B(LS2)} \quad (6b)$$

$$m_T (\rho_{CM} + \lambda_C) \ddot{\theta}_{C2} + m_T \rho_{CM} \ddot{\theta}_{G2} = - (W_T + m_T a_{B(LS1)}) \theta_{C2} - m_T a_{B(LS3)} \quad (6c)$$

Equation (6a) gives an expression for the force in the cable. Equations (6b) and (6c) are equations of motion containing Euler angles and second time-derivatives of the Euler angles.

The next step is to sum the moments about the center-of-mass of the P body and equate this to the time derivative of the angular momentum,

$$\Sigma \bar{M}_{CM(P)} = \dot{\bar{H}}_{CM(P)} \quad (7)$$

The time derivative of the angular momentum is given by the expression

$$\dot{\bar{H}}_{CM(P)} = I_P \dot{\bar{\omega}}_P + \bar{\omega}_P \times (I_P \bar{\omega}_P) \quad (8)$$

where I_P is the inertia tensor for the P body and $\bar{\omega}_P$ is its angular velocity. The term $\bar{\omega}_P \times (I_P \bar{\omega}_P)$ consists entirely of second-order quantities and can be neglected since higher-order terms are assumed to be negligible. The sum of the moments acting on body P is the torque applied to the instrument by the gimbal torque motors.

$$\Sigma \bar{M}_{CM(P)} = \bar{T}_{P/G} \quad (9)$$

To express $\bar{T}_{P/G}$ in terms of the gimbal axis torques, the transformation from the G coordinate system to the P coordinate system must be considered. Appendix A shows how $\bar{T}_{P/G}$ is expressed in terms of the gimbal torques. Substituting $\dot{\bar{H}}_{CM(P)} = I_P \dot{\bar{\omega}}_P$ and the expression for $\bar{T}_{P/G}$ from Appendix A into equation (7), the following three equations of motion are obtained.

$$I_{p11} \dot{\omega}_{p1} + I_{p12} \dot{\omega}_{p2} + I_{p13} \dot{\omega}_{p3} = T_A' \cos \theta_{CE} - T_R' \sin \theta_{CE} \quad (10a)$$

$$I_{p21} \dot{\omega}_{p1} + I_{p22} \dot{\omega}_{p2} + I_{p23} \dot{\omega}_{p3} = T_E' \quad (10b)$$

$$I_{p31} \dot{\omega}_{p1} + I_{p32} \dot{\omega}_{p2} + I_{p33} \dot{\omega}_{p3} = T_A' \sin \theta_{CE} + T_R' \cos \theta_{CE} \quad (10c)$$

The above equations do not contain any Euler angles or their time derivatives explicitly, but they could be obtained through the relationship between $\bar{\omega}_P$ and the Euler angle rates. The equations are retained in the above form because the rates ω_{p1} , ω_{p2} , and ω_{p3} will later be used as state variables. They are quantities that are measured by the rate gyros.

The third step in deriving the equations of motion is to solve the vector equation

$$\Sigma \bar{M}_{CM(G)} = \dot{\bar{H}}_{CM(G)} \quad (11)$$

As for the P body, $\dot{\bar{H}}_{CM(G)} = I_G \dot{\omega}_G$. The moments acting on the G body are given by

$$\Sigma \bar{M}_{CM(G)} = - \bar{\rho}_G \times \bar{f}_C - (\bar{\rho}_G - \bar{\rho}_P) \times \bar{f}_{G/P} + \bar{T}_{G/P} + \bar{T}_{G/C} \quad (12)$$

The torque $\bar{T}_{G/P}$ is the torque applied to the gondola by the gimbal torquers. It is equal to the negative of $\bar{T}_{P/G}$ derived in Appendix A. $\bar{T}_{G/C}$ is the torque applied to body G due to the torsion in the cable. $\bar{T}_{G/C}$ is derived in Appendix B. The force $\bar{f}_{G/P}$ is the interconnection force between the G body and the P body. When summing moments on body P, this force did not cause a moment because the center-of-mass of the P body is at the center of the gimbals, which is the point of application of the force $\bar{f}_{G/P}$. This force does, however, cause a moment about the center-of-mass of the gondola. By transforming each element of equation (12) into the G coordinate frame and substituting these results into equation (11), three more equations of motion can be obtained. However, one element in equation (12) is still unknown. The force $\bar{f}_{G/P}$ must be determined. By solving a fourth basic equation, the force $\bar{f}_{P/G}$, which is the negative of $\bar{f}_{G/P}$, can be found. Solving the equation

$$\Sigma \bar{F}(P) = m_P (\ddot{\bar{r}}_P + \bar{a}_B) \quad (13)$$

will yield the force $\bar{f}_{P/G}$. The sum of the forces on body P can be expressed as

$$\Sigma \bar{F}(P) = \bar{f}_{P/G} + \bar{f}_{P(g)} \quad (14)$$

where $\bar{f}_{P(g)}$ is simply the weight of body P. As with equation (1), all the elements in equation (13) are transformed to the LS frame. This results in three equations that give the three components of the force $\bar{f}_{P/G}$, which is the negative of $\bar{f}_{G/P}$ needed for equation (12). These expressions are transformed to the G coordinate frame where they are substituted into equation (12). The resulting equations of motion from equation (11) are

$$I_{G11} \dot{\omega}_{G1} + I_{G12} \dot{\omega}_{G2} + I_{G13} \dot{\omega}_{G3} = - T_A' - K_{G/C} \theta_{G1} \quad (15a)$$

$$\begin{aligned}
I_{G21} \dot{\omega}_{G1} + I_{G22} \dot{\omega}_{G2} + I_{G23} \dot{\omega}_{G3} = & - T_E' - \rho_G m_T (g + a_{B(LS1)}) \theta_{G2} \\
& + (\rho_G - \rho_P) [(\theta_{G2} + \theta_{C2}) (W_P + m_P a_{B(LS1)})] \\
& + m_P (\ell_C \ddot{\theta}_{C2} + a_{B(LS3)} + m_P \rho_P (\ddot{\theta}_{G2} + \ddot{\theta}_{C2}))
\end{aligned} \tag{15b}$$

$$\begin{aligned}
I_{G31} \dot{\omega}_{G1} + I_{G32} \dot{\omega}_{G2} + I_{G33} \dot{\omega}_{G3} = & - T_R' - \rho_G m_T (g + a_{B(LS1)}) \theta_{G3} \\
& - (\rho_G - \rho_P) [-(\theta_{G3} + \theta_{C3}) (W_P + m_P a_{B(LS1)})] \\
& + m_P (a_{B(LS2)} - \ell_C \ddot{\theta}_{C3}) - m_P \rho_P (\ddot{\theta}_{G3} + \ddot{\theta}_{C3})
\end{aligned} \tag{15c}$$

The above equations contain the components of the angular rate vector $\bar{\omega}_G$. These components can be eliminated from the equations by expressing them in terms of the Euler angle rates. Appendix C shows how this substitution is derived.

DEFINING THE STATE VARIABLES

Most of the equations of motion that were derived are second-order differential equations. These equations can be transformed to first-order equations by defining eight additional state variables. Let Ω denote the time rate of change of an Euler angle, or

$$\Omega = \dot{\theta} \tag{16}$$

Then the time derivative of Ω will be

$$\dot{\Omega} = \ddot{\theta} \tag{17}$$

and the substitution can be used to transform the equations of motion to first-order form. Equation (16) will give additional state equations needed to find the time-histories of the Euler angles θ_{C2} , θ_{C3} , θ_{G1} , θ_{G2} , and θ_{G3} .

Since $\bar{\omega}_P$ was not eliminated from equations (10a), (10b), and (10c), equation (16) cannot be used to express the time histories of θ_{P1} , θ_{P2} , and θ_{P3} . The relationship between $\bar{\omega}_P$ and the time derivatives of these three Euler angles must be derived. The details of developing this relationship are explained in Reference 3. The final three state equations that result from this relationship are

$$\dot{\theta}_{P1} = \omega_{P1} \cos \theta_{CE} + \omega_{P3} \sin \theta_{CE} - \Omega_{G1} - \Omega_{B1} \quad (18)$$

$$\dot{\theta}_{P2} = \omega_{P2} - \Omega_{G2} - \Omega_{C2} \quad (19)$$

$$\dot{\theta}_{P3} = -\omega_{P1} \sin \theta_{CE} + \omega_{P3} \cos \theta_{CE} - \Omega_{G3} - \Omega_{C3} \quad (20)$$

The other 13 state equations previously derived, with all substitutions and simplifications included, are given below.

$$-m_T (\rho_{CM} + \rho_C) \dot{\Omega}_{C3} - m_T \rho_{CM} \dot{\Omega}_{G3} = (W_T + m_T a_{B(LS1)}) \theta_{C3} - m_T a_{B(LS2)} \quad (21)$$

$$m_T (\rho_C + \rho_{CM}) \dot{\Omega}_{C2} + m_T \rho_{CM} \dot{\Omega}_{G2} = - (W_T + m_T a_{B(LS1)}) \theta_{C2} - m_T a_{B(LS3)} \quad (22)$$

$$I_{P11} \dot{\omega}_{P1} + I_{P12} \dot{\omega}_{P2} + I_{P13} \dot{\omega}_{P3} = T_A' \cos \theta_{CE} - T_R' \sin \theta_{CE} \quad (23)$$

$$I_{P21} \dot{\omega}_{P1} + I_{P22} \dot{\omega}_{P2} + I_{P23} \dot{\omega}_{P3} = T_E' \quad (24)$$

$$I_{P31} \dot{\omega}_{P1} + I_{P32} \dot{\omega}_{P2} + I_{P33} \dot{\omega}_{P3} = T_A' \sin \theta_{CE} + T_R' \cos \theta_{CE} \quad (25)$$

$$I_{G11} \dot{\Omega}_{G1} + I_{G12} \dot{\Omega}_{G2} + I_{G12} \dot{\Omega}_{C2} + I_{G13} \dot{\Omega}_{G3} + I_{G13} \dot{\Omega}_{C3} = -T_A' - K_{G/C} \theta_{G1} - I_{G11} \dot{\Omega}_{B1} \quad (26)$$

$$\begin{aligned}
& I_{G21} \dot{\Omega}_{G1} + (I_{G22} - \rho_{G-P} m_P \rho_P) \dot{\Omega}_{G2} + [I_{G22} - \rho_{G-P} m_P (\ell_C + \rho_P)] \dot{\Omega}_{C2} + I_{G23} \dot{\Omega}_{G3} \\
& + I_{G23} \dot{\Omega}_{C3} = - T_E' + [\rho_{G-P} (W_P + m_P a_{B(LS1)})] \theta_{C2} + \rho_{G-P} m_P a_{B(LS3)} \\
& - I_{G21} \dot{\Omega}_{B1} + [-\rho_G W_T - \rho_G m_T a_{B(LS1)}] \\
& + \rho_{G-P} (W_P + m_P a_{B(LS1)}) \theta_{G2} \tag{27}
\end{aligned}$$

$$\begin{aligned}
& I_{G31} \dot{\Omega}_{G1} + I_{G32} \dot{\Omega}_{G2} + I_{G32} \dot{\Omega}_{C2} + [I_{G33} - \rho_{G-P} m_P \rho_P] \dot{\Omega}_{G3} \\
& + [I_{G33} - \rho_{G-P} m_P (\rho_P + \ell_C)] \dot{\Omega}_{C3} = - T_R' + \rho_{G-P} (W_P + m_P a_{B(LS1)}) \theta_{C3} \\
& - \rho_{G-P} m_P a_{B(LS2)} - I_{G31} \dot{\Omega}_{B1} + [\rho_{G-P} (W_P + m_P a_{B(LS1)})] \\
& - \rho_G (W_T + m_T a_{B(LS1)}) \theta_{G3} \tag{28}
\end{aligned}$$

$$\dot{\theta}_{C2} = \Omega_{C2} \tag{29}$$

$$\dot{\theta}_{C3} = \Omega_{C3} \tag{30}$$

$$\dot{\theta}_{G1} = \Omega_{G1} \tag{31}$$

$$\dot{\theta}_{G2} = \Omega_{G2} \tag{32}$$

$$\dot{\theta}_{G3} = \Omega_{G3} \tag{33}$$

The eight second-order differential equations have now been replaced by 16 first-order differential equations. Using matrix notation, these equations are in the form

$$M \dot{\bar{X}} = K \bar{X} + C \bar{U} \tag{34}$$

This can be easily transformed to the common state variable form,

$$\dot{\bar{X}} = A \bar{X} + B \bar{U} \quad , \quad (35)$$

by inverting the M matrix. This inversion is easily performed in the computer simulation, so it is not necessary to perform it analytically.

The state vector \bar{X} is defined as

$$\bar{X} = \left\{ \begin{array}{l} \Omega_{C2} \\ \Omega_{C3} \\ \Omega_{G1} \\ \Omega_{G2} \\ \Omega_{G3} \\ \omega_{P1} \\ \omega_{P2} \\ \omega_{P3} \\ \theta_{C2} \\ \theta_{C3} \\ \theta_{G1} \\ \theta_{G2} \\ \theta_{G3} \\ \theta_{P1} \\ \theta_{P2} \\ \theta_{P3} \end{array} \right\} \quad (36)$$

It contains the eight Euler angles that were originally sought to define the position of the system. The Euler rates of the C and G coordinate transformations, and the body rates of the instrument, are also state variables.

The input vector \bar{U} is defined as

$$\bar{U} = \left\{ \begin{array}{c} T_A' \\ T_E' \\ T_R' \\ a_{B(LS2)} \\ a_{B(LS3)} \\ \dot{\Omega}_{B1} \\ \Omega_{B1} \end{array} \right\} \quad (37)$$

which contains the gimbal torques and the balloon motions.

Notice that the A matrix in equation (35) will be a function of time whenever the balloon has a nonconstant acceleration. This is because the component $a_{B(LS1)}$ is present in the A matrix.

Until now, the azimuth bearing friction and the stiffness of the flex pivots have been neglected. Modeling these parameters will now be considered.

MODELING THE GIMBAL BEARINGS

In the present effort, a very simple model is used for the gimbal bearings. This model is sufficient for the preliminary control system design, but a more accurate model may be desired if a more precise analysis is needed.

The azimuth bearing is modeled using a Dahl friction model, which is explained in detail in Reference 2. Using small angle assumptions and neglecting higher-order terms, the following equation for the resistance torque of the bearing results.

$$\dot{T}_{fA} = \gamma T_{fo}^2 \dot{\theta}_{P1} \quad (38)$$

The parameters γ and T_{fo} are constants that depend on the physical properties of the bearing. Using values given in Reference 2 as guidelines, values are chosen for γ and T_{fo} that will yield reasonable values for T_{fA} relative to the azimuth gimbal torque, T_A . Again, for a more precise analysis, these values would have to be determined for the particular azimuth bearing.

Equation (38) can be integrated directly to give the resistance torque

$$T_{fA} = J_A \theta_{P1} \quad (39)$$

where J_A is a constant,

$$J_A = \gamma T_{fo}^2 .$$

The flex pivot bearings act as torsional springs and are, therefore, modeled as linear torsional springs with a resistance torque given by

$$T_{res} = J \theta \tag{40}$$

where J is the spring constant. Values for J are available from Reference 1. The resistance torques of the two flex pivots are expressed as

$$T_{fE} = J_E \theta_{P2} \tag{41}$$

$$T_{fR} = J_R \theta_{P3} \tag{42}$$

for the fine elevation and roll axes, respectively. To include the bearing models into equations (18) through (33), the following substitutions are made in those equations

$$T_A' = T_A + J_A \theta_{P1} \tag{43}$$

$$T_E' = T_E + J_E \theta_{P2} \tag{44}$$

$$T_R' = T_R + J_R \theta_{P3} \tag{45}$$

where T_A , T_E , and T_R now represent the actual torque applied by the torque motors on the gimbal system.

This completes the development of the equations of motion. The equations can now be used to simulate the motion of the system for the various inputs in the \bar{U} vector. The next step is to develop a control law to point the instrument at the Sun. This control law will determine the gimbal axis torques required to point the instrument.

DEVELOPING THE CONTROL LAW

Since the objective is to point the instrument at the Sun, a relationship between the S reference frame and the P reference frame must be developed. If it is assumed that azimuth and coarse elevation axes motions have pointed the instrument relatively close to the Sun, the small angle assumption can be used when defining a transformation between the S and the P reference frames. Using this assumption and neglecting the second-order terms, an Euler transformation can be developed.

$$\begin{Bmatrix} \bar{i}_{P1} \\ \bar{i}_{P2} \\ \bar{i}_{P3} \end{Bmatrix} = \begin{bmatrix} 1 & \theta_3 & -\theta_2 \\ -\theta_3 & 1 & \theta_1 \\ \theta_2 & -\theta_1 & 1 \end{bmatrix} \begin{Bmatrix} \bar{i}_{S1} \\ \bar{i}_{S2} \\ \bar{i}_{S3} \end{Bmatrix} \quad (46)$$

Another transformation matrix can be developed between the S and P frames by making successive Euler transformations from the S to LS, LS to B, B to C, C to G, and G to P coordinate frames. This transformation matrix will be a function of the Euler angles of the system that result from the equations of motion. By equating this matrix to that of equation (46), the angles θ_1 , θ_2 , and θ_3 can be determined in terms of the other Euler angles. The following expressions are found for these angles.

$$\theta_1 = (\theta_{B1} + \theta_{G1} + \theta_{P1}) \cos \theta_{CE} - (\theta_{C3} + \theta_{G3} + \theta_{P3}) \sin \theta_{CE} \quad (47)$$

$$\theta_2 = \theta_{C2} + \theta_{G2} + \theta_{P2} \quad (48)$$

$$\theta_3 = (\theta_{C3} + \theta_{G3} + \theta_{P3}) \cos \theta_{CE} + (\theta_{B1} + \theta_{G1} + \theta_{P1}) \sin \theta_{CE} \quad (49)$$

Angles θ_1 , θ_2 , and θ_3 are the Euler angles between the P and S reference frames. When these angles are zero, the two frames will be essentially colinear and the instrument will be essentially pointing at the Sun, because the distance to the Sun is much greater than the distance between the origins of the P and S reference frames. Now that the pointing angles θ_1 , θ_2 , and θ_3 have been determined in terms of the Euler angles of the system, a control law can be developed which will maintain these angles close to zero.

The control law for each axis is defined to be

$$T_{(Pi)com} = K_{Ri} [K_{Pi} (\theta_{(i)com} - \theta_i) - \omega_{Pi} + K_{Ii} \dot{\theta}_i] \quad (50)$$

The command angle $\theta_{(i)com}$ is included so that the instrument can be pointed at a source slightly different from the center of the Sun. Figure 3 shows a single axis block diagram of the control law along with a simplified model for the plant dynamics. The integral control is required to remove any steady-state offset in the response. The rate commands ω_{cI} and ω_{cP} must be limited due to the limited control torque $T_{(Pi)com}$. These limiters are shown in the block diagram. The limiter on ω_{cI} must be such that the integrator stops integrating while ω_{cI} is at its limit.

The commanded torques obtained from equation (50) are the commanded torques to be applied to the instrument along the axes of the P coordinate system (body axes). These commanded torques must be transformed as shown in Appendix A to determine the commanded gimbal torques $T_{(A)com}$, $T_{(E)com}$, and $T_{(R)com}$.

In the present analysis, the gains K_{Ii} , K_{Pi} , and K_{Ri} were chosen to yield a phase margin of 65 deg and a gain crossover frequency of 1 Hz for the system in Figure 3 when linearized.




Figure 3

SIMULATION RESULTS

A computer simulation of the system was written using equations (18) through (33) and the control scheme developed in the previous section. The sampling rate of the digital flight computer, 10 samples per second, was included. The sensors and torque motors were assumed to be ideal, except for a limit on the maximum torque available from the torque motors. A listing of the simulation code is included in Appendix D.

Simulation results are shown in Figures 4 through 8. These results were generated with zero initial conditions for all of the state variables. This means the instrument is initially pointing at the Sun. A step input for the command angles θ_1 , θ_2 , and θ_3 of 0.003 radians is introduced at time $t = 0$. A limit of 0.0006 rad/sec in each axis was placed on the rate command in the integral control path. A limit of this same value was also placed on the rate command in each axis in the proportional control path. These limits resulted in a desirable response. Figures 4, 5, and 6 show the gimbal axis torques and the response of the angles θ_1 , θ_2 , and θ_3 to the step input. The elevation and roll torques saturate at times. Figure 7 shows the components of the angular rate of the instrument.

Figure 8 shows the G_3 components of the pendulous motion of the gondola relative to the C coordinate frame. This behavior is typical of the cable motion relative to the balloon also.

CONCLUSIONS

The balloon-pointing-system was modeled as a two rigid-body system. Equations describing the motion of the system were derived and a control law was adopted to point the science instrument at the Sun. A simulation program was developed and results from the simulation indicate that the instrument can be pointed with favorable response characteristics.

Some simplifying assumptions were made during the derivation of these equations of motion. The equations, however, are sufficient for a practical analysis of the control system.





Figure 4





Figure 5





Figure 6






Figure 7





Figure 8

APPENDIX A

The commanded torques from the control law are commanded torques in the body axes of the instrument. That is,

$$\bar{T}_{P/G} = T_{P/G(P1)} \bar{i}_{P1} + T_{P/G(P2)} \bar{i}_{P2} + T_{P/G(P3)} \bar{i}_{P3} \quad (A-1)$$

The commanded gimbal torques, however, are commanded torques along the Euler axes of the transformation between the G and P frames. Four rotations are used to define this transformation: azimuth, fine elevation, fine roll, and coarse elevation. The transformation is expressed mathematically as

$$\begin{Bmatrix} \bar{i}_{P1} \\ \bar{i}_{P2} \\ \bar{i}_{P3} \end{Bmatrix} = \begin{bmatrix} \cos \theta_{CE} & 0 & -\sin \theta_{CE} \\ 0 & 1 & 0 \\ \sin \theta_{CE} & 0 & \cos \theta_{CE} \end{bmatrix} \begin{bmatrix} 1 & \theta_{P3} & 0 \\ -\theta_{P3} & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & -\theta_{P2} \\ 0 & 1 & 0 \\ \theta_{P2} & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & \theta_{P1} \\ 0 & -\theta_{P1} & 1 \end{bmatrix} \begin{Bmatrix} \bar{i}_{G1} \\ \bar{i}_{G2} \\ \bar{i}_{G3} \end{Bmatrix} \quad (A-2)$$

where small angles are assumed except in the case of the coarse elevation angle θ_{ce} .

The commanded torque on body P can be expressed in terms of the commanded gimbal torques for the azimuth, fine elevation, and fine roll axes as,

$$\bar{T}_{P/G} = T_A' \bar{i}_{G1} + T_E' \bar{i}_{P2}' + T_R' \bar{i}_{P3}'' \quad (A-3)$$

The coarse elevation gimbal is assumed to be mechanically locked in this development. The primes are put on T_A , T_E , and T_R here because the effects of the gimbal bearings have not yet been included in the model. This notation will avoid confusion later on. The unit vector \bar{i}_{P2}' is the unit vector along the new 2-axis that results after the rotation of θ_{P1} about the G_1 axis in equation (A-2). The unit vector \bar{i}_{P3}'' is the new 3-axis that results from the rotation through angle θ_{P2} . Using equation (A-2), the relationships between \bar{i}_{G1} , \bar{i}_{P2}' , \bar{i}_{P3}'' , and the unit vectors of the P reference frame can be determined. Equation (A-3) can then be expressed in terms of the unit vectors \bar{i}_{P1} , \bar{i}_{P2} , and \bar{i}_{P3} . Equating this result to equation (A-1) gives the relationship between the control torques and the gimbal axis torques. These relationships are, in matrix form,

$$\begin{Bmatrix} T_A' \\ T_E' \\ T_R' \end{Bmatrix} = \begin{bmatrix} \cos \theta_{CE} & -\theta_{P3} & \sin \theta_{CE} \\ \theta_{P3} \cos \theta_{CE} & 1 & \theta_{P3} \sin \theta_{CE} \\ -(\sin \theta_{CE} + \theta_{P2} \cos \theta_{CE}) & 0 & (\cos \theta_{CE} - \theta_{P2} \sin \theta_{CE}) \end{bmatrix} \begin{Bmatrix} T_{P/G(P1)} \\ T_{P/G(P2)} \\ T_{P/G(P3)} \end{Bmatrix}$$

(A-4)

APPENDIX B

The torsion in the cable can be expressed as

$$\bar{T}_{G/C} = - K_{G/C} \alpha \bar{i}_{C1} \quad , \quad (B-1)$$

where α is the angle of twist in the cable. For the small angle assumption, the angle α will be about the same as the Euler angle θ_{G1} ,

$$\alpha \approx \theta_{G1} \quad . \quad (B-2)$$

From the Euler transformation between the C and the G frames, \bar{i}_{C1} can be expressed in terms of the G frame unit vectors as,

$$\bar{i}_{C1} = \bar{i}_{G1} - \theta_{G3} \bar{i}_{G2} + \theta_{G2} \bar{i}_{G3} \quad (B-3)$$

assuming small angles for θ_{G2} and θ_{G3} . Substituting equations (B-2) and (B-3) into equation (B-1) and neglecting second-order terms, the cable torsion becomes

$$\bar{T}_{G/C} = - K_{G/C} \theta_{G1} \bar{i}_{G1} \quad . \quad (B-4)$$

APPENDIX C

The vector $\bar{\omega}_G$ is the angular rate of the gondola. Referring to Figure 2, this can be expressed as

$$\begin{aligned} \bar{\omega}_G = & \dot{\theta}_{B1} \bar{i}_{B1} + \dot{\theta}_{C2} \bar{i}_{B2} + \dot{\theta}_{C3} \bar{i}_{C3} + \dot{\theta}_{G3} \bar{i}_{C3} + \dot{\theta}_{G2} (\cos \theta_{G1} \bar{i}_{G2} - \sin \theta_{G1} \bar{i}_{G3}) \\ & + \dot{\theta}_{G1} \bar{i}_{G1} \end{aligned} \quad (C-1)$$

Let

$$\dot{\theta}_{B1} = \Omega_{B1} \quad (C-2)$$

$$\dot{\theta}_{C2} = \Omega_{C2} \quad (C-3)$$

$$\dot{\theta}_{C3} = \Omega_{C3} \quad (C-4)$$

$$\dot{\theta}_{G1} = \Omega_{G1} \quad (C-5)$$

$$\dot{\theta}_{G2} = \Omega_{G2} \quad (C-6)$$

$$\dot{\theta}_{G3} = \Omega_{G3} \quad (C-7)$$

Substituting equations (C-2) through (C-7) into equation (C-1), expressing the unit vectors in equation (C-1) in terms of the unit vectors ($\bar{i}_{G1}, \bar{i}_{G2}, \bar{i}_{G3}$), assuming $\sin \theta \doteq \theta$ and $\cos \theta \doteq 1$ for all the θ 's in the resulting equation for (C-1), and neglecting the higher order terms in the Ω 's, the θ 's and their product yields

$$\bar{\omega}_G = \omega_{G1} \bar{i}_{G1} + \omega_{G2} \bar{i}_{G2} + \omega_{G3} \bar{i}_{G3} \quad (C-8)$$

where

$$\omega_{G1} = \Omega_{G1} + \Omega_{B1} \quad (C-9)$$

$$\omega_{G2} = \Omega_{G2} + \Omega_{C2} \quad (C-10)$$

$$\omega_{G3} = \Omega_{G3} + \Omega_{C3} \quad (C-11)$$

Taking the time derivative of equation (C-8) and neglecting higher order terms in ω_{Gi} , $i = 1, 2, 3$ and their product produces

$$\dot{\omega}_G = \dot{\omega}_{G1} \bar{i}_{G1} + \dot{\omega}_{G2} \bar{i}_{G2} + \dot{\omega}_{G3} \bar{i}_{G3} \quad (C-12)$$

where

$$\dot{\omega}_{G1} = \dot{\Omega}_{G1} + \dot{\Omega}_{B1} \quad (C-13)$$

$$\dot{\omega}_{G2} = \dot{\Omega}_{G2} + \dot{\Omega}_{C2} \quad (C-14)$$

$$\dot{\omega}_{G3} = \dot{\Omega}_{G3} + \dot{\Omega}_{C3} \quad (C-15)$$

Equations (C-13) through (C-15) follow from equation (C-9) through (C-11).

APPENDIX D. LISTING OF COMPUTER PROGRAM

```

C*****
C PROGRAM TO SIMULATE P.E.'S PROPOSED GONDOLA CONCEPT AND          +
C GIMBAL SYSTEM FOR THE BALLOON POINTING SYSTEM                      +
C THIS CODE CONTAINS THE SIMULATION OF THE EOM'S WITH THE           +
C CONTROL LAW, INTEGRAL CONTROL, AND THE SAMPLE & HOLD OF THE      +
C DIGITAL COMPUTER INCLUDED. ALSO, LIMITS ARE IMPOSED ON THE       +
C TORQUE OF THE GIMBAL TORQUERS, THE RATE COMMAND DUE TO THE      +
C INTEGRAL CONTROL, AND THE RATE COMMAND DUE TO THE ANGLE ERROR.   +
C IMSL ROUTINES ARE USED FOR MATRIX INVERSION, MATRIX              +
C MULTIPLICATION, AND INTEGRATION OF THE STATE EQUATIONS.         +
C                               BILLY LIGHTSEY                       +
C   POINTING CONTROL SYSTEMS BRANCH/CONTROL SYSTEMS DIVISION      +
C                               APRIL 1988                          +
C*****
      IMPLICIT REAL(I,K)
      REAL M(16,16),A(16,16),B(16,7),X(16),U(7),Z(16,16),MT,LC,MP,MG,
&I1,I2,I3,IM1,IM2,IM3,I1MAX,I2MAX,I3MAX
      DIMENSION PARAM(50)
      INTEGER IDO
      EXTERNAL FCN,IVPRK
      COMMON /SUB1/ A,B,Z,WT,MT,RG, RGP,WP,U,AB1,MP
      COMMON /SUB2/ AB10,AB11,AB20,AB21,AB30,AB31,OMB0,OMB1,OMGAB,TB10,
&TB1
      COMMON /TIMES/ TAB11,TAB12,TAB21,TAB22,TAB31,TAB32,TOMB1,TOMB2
      NAMELIST /INPUT/ LC,RP,WG,KGC,RG,WP,TCE,IP11,IP12,IP13,
&IP21,IP22,IP23,IP31,IP32,IP33,IG11,IG12,IG13,IG21,IG22,IG23,IG31,
&IG32,IG33,AB10,AB11,AB20,AB21,AB30,AB31,OMB0,OMB1,OMGAB,TB10,
&DELT,TI,TF,TAB11,TAB12,TAB21,WC,PM1,PM2,PM3,IPRT,X,
&TAB22,TAB31,TAB32,TOMB1,TOMB2,TFA,TFE,TFR,TC1,TC2,TC3,TQMAX,
&WCT1,WCT2,WCT3,WCI1,WCI2,WCI3
      READ(5,INPUT)
      MG=WG/32.2
      MP=WP/32.2
      MT=MG+MP
      WT=WG+WP
      CTCE=COS(TCE)
      STCE=SIN(TCE)
      RCM=(MP*RP+MG*RG)/MT
      RGP=RG-RP
C CONVERT PHASE MARGINS TO RADIANES
      PM1=PM1*3.14159266/180.
      PM2=PM2*3.14159266/180.
      PM3=PM3*3.14159266/180.
C COMPUTE CORNER FREQUENCIES TO CORRESPOND TO THE PHASE MARGINS
      WC2=WC**2
      TNP=TAN(PM1)
      W11=WC*(SQRT(TNP*TNP+1)-TNP)
      TNP=TAN(PM2)
      W12=WC*(SQRT(TNP*TNP+1)-TNP)
      TNP=TAN(PM3)
      W13=WC*(SQRT(TNP*TNP+1)-TNP)
      W21=WC2/W11
      W22=WC2/W12
      W23=WC2/W13
C COMPUTE THE GAINS KI, KP, KR
      KI1=(WC2*SQRT(1+WC2/W21**2))/SQRT(1+WC2/W11**2)
      KI2=(WC2*SQRT(1+WC2/W22**2))/SQRT(1+WC2/W12**2)
      KI3=(WC2*SQRT(1+WC2/W23**2))/SQRT(1+WC2/W13**2)
      KP1=KI1/W11
      KP2=KI2/W12

```

```

      KP3=KI3/W13
      KR1=W21*IP11
      KR2=W22*IP22
      KR3=W23*IP33
C INITIALIZE M,A,AND B MATRICES TO ZERO
      DO 1 N=1,16
      DO 2 J=1,16
      M(N,J)=0.0
2     A(N,J)=0.0
      DO 3 L=1,7
3     B(N,L)=0.0
1     CONTINUE
C CREATE THE M MATRIX
      M(1,2)=-MT*(RCM+LC)
      M(1,5)=-MT*RCM
      M(2,1)=-MT*(LC+RCH)
      M(2,4)=-MT*RCM
      M(3,6)=-IP11
      M(3,7)=-IP12
      M(3,8)=-IP13
      M(4,6)=-IP21
      M(4,7)=-IP22
      M(4,8)=-IP23
      M(5,6)=-IP31
      M(5,7)=-IP32
      M(5,8)=-IP33
      M(6,3)=-IG11
      M(6,4)=-IG12
      M(6,5)=-IG13
      M(6,1)=-IG12
      M(6,2)=-IG13
      M(7,3)=-IG21
      M(7,4)=-IG22-RGP*MP*RP
      M(7,5)=-IG23
      M(7,1)=-IG22-RGP*MP*(LC+RP)
      M(7,2)=-IG23
      M(8,3)=-IG31
      M(8,4)=-IG32
      M(8,5)=-IG33-RGP*MP*RP
      M(8,1)=-IG32
      M(8,2)=-IG33-RGP*MP*(RP+LC)
      DO 4 N=9,16
4     M(N,N)=1.0
C INVERT THE M MATRIX
      CALL LINRG(16,M,16,Z,16)
C CREATE NEEDED ELEMENTS OF K MATRIX.THE TIME-DEPENDENT ELEMENTS ARE
C CALCULATED IN SUBROUTINE FCN
      K611=-KGC
      K146=CTCE
      K148=STCE
      K166=-STCE
      K168=CTCE
      K314=-TFA*CTCE
      K316=TFR*STCE
      K415=-TFE
      K514=-TFA*STCE
      K516=-TFR*CTCE
      K614=TFA
      K715=TFE
      K816=TFR

```



```

C CREATE THE NEEDED ELEMENTS OF C MATRIX
  C14=-MT
  C25=-MT
  C31=CTCE
  C33=-STCE
  C66=-IG11
  C76=-IG21
  C51=STCE
  C53=CTCE
  C86=-IG31
  C75=RGF*MP
  C84=-RGF*MP
C GET THE A AND B MATRICIES.SINCE K AND C CONTAIN MOSTLY ZEROS,IT IS
C BENEFICIAL TO MULTIPLY ONLY THE NECESSARY ELEMENTS INSTEAD OF THE
C ENTIRE MATRICIES
  DO 5 J=1,16
    A(J,1)=-Z(J,9)-Z(J,15)
    A(J,2)=-Z(J,10)-Z(J,16)
    A(J,3)=-Z(J,11)-Z(J,14)
    A(J,4)=-Z(J,12)-Z(J,15)
    A(J,5)=-Z(J,13)-Z(J,16)
    A(J,6)=-Z(J,14)*K146+Z(J,16)*K166
    A(J,7)=-Z(J,15)
    A(J,8)=-Z(J,14)*K148+Z(J,16)*K168
    A(J,11)=-Z(J,6)*K611
    A(J,14)=-Z(J,3)*K314+Z(J,5)*K514+Z(J,6)*K614
    A(J,15)=-Z(J,4)*K415+Z(J,7)*K715
    A(J,16)=-Z(J,3)*K316+Z(J,5)*K516+Z(J,8)*K816
    B(J,1)=-Z(J,3)*C31+Z(J,5)*C51-Z(J,6)
    B(J,2)=-Z(J,4)-Z(J,7)
    B(J,3)=-Z(J,3)*C33+Z(J,5)*C53-Z(J,8)
    B(J,4)=-Z(J,1)*C14+Z(J,8)*C84
    B(J,5)=-Z(J,2)*C25+Z(J,7)*C75
    B(J,6)=-Z(J,6)*C66+Z(J,7)*C76+Z(J,8)*C86
  5 B(J,7)=-Z(J,14)
C SET UP THE INTEGRATION PROCEDURE
  T=TI
  TOL=.0005
  IDO=1
C SET UP THE PARAM ARRAY REQUIRED BY IMSL ROUTINE IVPK
  DO 6 J=1,50
  6 PARAM(J)=0.0
  PARAM(4)=3000.0
  PARAM(10)=1.0
C GET INITIAL VALUES OF INPUT VECTOR U FOR PRINTING AT TIME=0
  CALL UCALC(T,U,AB1,K)
C SET I.C.'S OF INTEGRAL CONTROL PARAMETER I
  I1=0.0
  I2=0.0
  I3=0.0
  IPT=IPRT
C SET LIMITS ON INTEGRAL AND ANGLE RATE COMMANDS
  I1MAX=WCI1/KI1
  I2MAX=WCI2/KI2
  I3MAX=WCI3/KI3
  TE1MAX=WCT1/KP1
  TE2MAX=WCT2/KP2
  TE3MAX=WCT3/KP3
C BEGIN ITERATION LOOP FOR INTEGRATION
  7 TEND=T+DELT

```

```

IM1=I1
IM2=I2
IM3=I3
THETA1=CTCE*(TB1+X(11)+X(14))-STCE*(X(10)+X(13)+X(16))
THETA2=X(9)+X(12)+X(15)
THETA3=CTCE*(X(10)+X(13)+X(16))+STCE*(TB1+X(11)+X(14))
IF(T.EQ.TO) GOTO 9
C IMPOSE LIMIT ON RATE COMMAND DUE TO ANGLE ERROR
TERR1=TC1-THETA1
TERR2=TC2-THETA2
TERR3=TC3-THETA3
ATERR1=ABS(TERR1)
ATERR2=ABS(TERR2)
ATERR3=ABS(TERR3)
IF(ATERR1.GT.TE1MAX) TERR1=TE1MAX*SIGN(1.,TERR1)
IF(ATERR2.GT.TE2MAX) TERR2=TE2MAX*SIGN(1.,TERR2)
IF(ATERR3.GT.TE3MAX) TERR3=TE3MAX*SIGN(1.,TERR3)
C IMPOSE LIMIT ON RATE COMMAND DUE TO INTEGRAL CONTROL
I1=IM1+DELT*TERR1
I2=IM2+DELT*TERR2
I3=IM3+DELT*TERR3
AI1=ABS(I1)
AI2=ABS(I2)
AI3=ABS(I3)
IF(AI1.GT.I1MAX) I1=I1MAX*SIGN(1.,I1)
IF(AI2.GT.I2MAX) I2=I2MAX*SIGN(1.,I2)
IF(AI3.GT.I3MAX) I3=I3MAX*SIGN(1.,I3)
9 CONTINUE
C NOW FIND THE CONTROL TORQUES
TP1=KR1*(KI1*I1+KP1*TERR1-X(6))
TP2=KR2*(KI2*I2+KP2*TERR2-X(7))
TP3=KR3*(KI3*I3+KP3*TERR3-X(8))
C CONVERT CONTROL TORQUES TO GIMBAL AXIS TORQUES TA,TE,AND TR
U(1)=CTCE*TP1-X(16)*TP2+STCE*TP3
U(2)=X(16)*CTCE*TP1+TP2+X(16)*STCE*TP3
U(3)=- (STCE+X(15)*CTCE)*TP1+(CTCE-X(15)*STCE)*TP3
C IMPOSE TORQUE LIMIT OF TQMAX FT-LBS
AU1=ABS(U(1))
IF(AU1.GT.TQMAX) U(1)=TQMAX*SIGN(1.,U(1))
AU2=ABS(U(2))
IF(AU2.GT.TQMAX) U(2)=TQMAX*SIGN(1.,U(2))
AU3=ABS(U(3))
IF(AU3.GT.TQMAX) U(3)=TQMAX*SIGN(1.,U(3))
IF(IPT.EQ.IPRT) THEN
WRITE(7) T,X,U,AB1,THETA1,THETA2,THETA3
IPT=0
ENDIF
IPT=IPT+1
C CALL IMSL ROUTINE TO INTEGRATE
CALL IVPRK(IDO,16,FCN,T,TEND,TOL,PARAM,X)
IF(T.GE.TF) STOP
C SET LIMIT ON INTEGRATION STEPS TO AVOID CHANCE OF AN INFINITE LOOP
IF(T.GE.3600) STOP
IF(T.LT.0.0) STOP
GOTO 7
END
-----
C SUBROUTINE TO CALCULATE THE R.H.S. OF THE DIFFERENTIAL EQUATIONS
SUBROUTINE FCN(N,T,X,DX)
IMPLICIT REAL(K)

```

```

      REAL X(16),DX(16),U(7),A(16,16),Z(16,16),B(16,7),AX(16),BU(16),
      &MT,MP
      COMMON /SUB1/ A,B,Z,WT,MT,RG,RGP,WP,U,AB1,MP
C CALCULATE THE TIME-DEPENDENT INPUTS
      CALL UCALC(T,U,AB1,X)
C CALCULATE TIME DEPENDENT ELEMENTS IN K AND A MATRICES
      K110=WT+MT*AB1
      K29=-WT-MT*AB1
      K712=-RG*(WT+MT*AB1)+RGP*(WP+MP*AB1)
      K813=K712
      K79=RGP*(WP+MP*AB1)
      K810=K79
      DO 1 J=1,16
      A(J,9)=Z(J,2)*K29+Z(J,7)*K79
      A(J,10)=Z(J,1)*K110+Z(J,8)*K810
      A(J,12)=Z(J,7)*K712
      1 A(J,13)=Z(J,8)*K813
C MULTIPLY A*X AND B*U USING IMSL ROUTINE MURRV
      CALL MURRV(16,16,A,16,16,X,1,16,AX)
      CALL MURRV(16,7,B,16,7,U,1,16,BU)
      DO 2 J=1,16
      2 DX(J)=AX(J)+BU(J)
      RETURN
      END
-----
C SUBROUTINE TO CALCULATE TIME DEPENDENT INPUTS
      SUBROUTINE UCALC(T,U,AB1,X)
      IMPLICIT REAL (K)
      COMMON /SUB2/ AB10,AB11,AB20,AB21,AB30,AB31,OMBO,OMB1,OMGAB,TB10,
      &TB1
      COMMON /TIMES/ TAB11,TAB12,TAB21,TAB22,TAB31,TAB32,TOMB1,TOMB2
      REAL U(7),X(16)
C BALLOON ACCELERATIONS ARE ZERO EXCEPT DURING TIME INTERVAL
C DEFINED BY TAB1 AND TAB2
      AB1=0.0
      TB1=0.0
      DO 2 I=4,7
      U(I)=0.0
      2 CONTINUE
      IF(T.GE.TAB11.AND.T.LE.TAB12) AB1=AB10+AB11*T
      IF(T.GE.TAB21.AND.T.LE.TAB22) U(4)=AB20+AB21*T
      IF(T.GE.TAB31.AND.T.LE.TAB32) U(5)=AB30+AB31*T
C ANGULAR MOTION OF BALLOON
      IF(T.LT.TOMB1.OR.T.GT.TOMB2) GOTO 5
      U(6)=OMBO+OMB1*T
      U(7)=OMBO*T+.5*T*T*OMB1+OMGAB
      TB1=.5*OMBO*T**2+T**3*OMB1/6.0+OMGAB*T+TB10
      5 RETURN
      END

```

DATA FILE

```
&INPUT
TI=0.0,
TF=15.,
TC1=0.003,
TC2=0.003,
TC3=0.003,
WCT1=.000645,
WCT2=.000645,
WCT3=.000645,
WCI1=.000645,
WCI2=.000645,
WCI3=.000645,
TQMAX=10.,
DELT=0.05,
IPRT=2,
TCE=0.500,
WC=6.28,
PM1=65.0,PM2=65.0,PM3=65.0,
KGC=1.0E02,
RP=15.0,
RG=25.0,
LC=350.0,
WG=2205.0,
WP=1765.0,
X=16*0.0,
IP11=79.0,
IP22=1830.0,
IP33=1830.0,
IP12=0.,IP13=0.,IP21=0.,IP23=0.,IP31=0.,IP32=0.,
IG11=3425.0,
IG22=1713.0,
IG33=1713.0,
IG12=0.,IG13=0.,IG21=0.,IG23=0.,IG31=0.,IG32=0.,
AB10=0.0,AB11=0.0,
AB20=0.0,AB21=0.0,
AB30=0.0,AB31=0.0,
OMB0=0.0,OMB1=0.0,
OMGAB=0.0,
TB10=0.0,
TAB11=0.0,TAB12=0.0,
TAB21=0.0,TAB22=0.0,
TAB31=0.0,TAB32=0.0,
TFA=11.0,
TFE=4.5,
TFR=4.5,
&END
```

REFERENCES

1. Perkin-Elmer Space Sciences Division: Stabilized Balloon Platform Study Final Report, Volume 1. February 1988.
2. Nurre, G. S.: An Analysis of the Dahl Friction Model and Its Effect on a CMG Gimbal Rate Controller. NASA TM X-64934, November 1974.
3. Burdeshaw, D. H.: Methods of Computing the Transformation Matrix Associated With Gimballess Inertial Measurement Units. NASA TM X-53294, July 1965.

APPROVAL

**MODELING, DESIGNING, AND SIMULATING A POINTING CONTROL
SYSTEM FOR BALLOON-BORNE SOLAR EXPERIMENTS**

By W. D. Lightsey and M. E. Polites

The information in this report has been reviewed for technical content. Review of any information concerning Department of Defense or nuclear energy activities or programs has been made by the MSFC Security Classification Officer. This report, in its entirety, has been determined to be unclassified.



G. F. McDONOUGH

Director, Structures and Dynamics Laboratory

1. REPORT NO. NASA TM-100337		2. GOVERNMENT ACCESSION NO.		3. RECIPIENT'S CATALOG NO.	
4. TITLE AND SUBTITLE Modeling, Designing, and Simulating a Pointing Control System for Balloon-Borne Solar Experiments				5. REPORT DATE July 1988	
				6. PERFORMING ORGANIZATION CODE	
7. AUTHOR(S) W. D. Lightsey and M. E. Polites				8. PERFORMING ORGANIZATION REPORT #	
9. PERFORMING ORGANIZATION NAME AND ADDRESS George C. Marshall Space Flight Center Marshall Space Flight Center, Alabama 35812				10. WORK UNIT NO.	
				11. CONTRACT OR GRANT NO.	
12. SPONSORING AGENCY NAME AND ADDRESS National Aeronautics and Space Administration Washington, D.C. 20546				13. TYPE OF REPORT & PERIOD COVERED Technical Memorandum	
				14. SPONSORING AGENCY CODE	
15. SUPPLEMENTARY NOTES Prepared by Structures and Dynamics Laboratory, Science and Engineering Directorate.					
16. ABSTRACT A simplified model of a pointing control system for balloon-borne solar experiments is chosen. Equations of motion for this model are derived and a feedback control law is defined. A digital computer simulation of the system is developed. Simulation results show favorable system response characteristics.					
17. KEY WORDS Pointing Control System Balloon-Borne Experiments Solar Experiments				18. DISTRIBUTION STATEMENT Unclassified - Unlimited	
19. SECURITY CLASSIF. (of this report) Unclassified		20. SECURITY CLASSIF. (of this page) Unclassified		21. NO. OF PAGES 39	22. PRICE NTIS